6. 某地区有三个农场共用一条灌渠，每个农场的可灌溉地及分配到的最大用水量如下表：

农场	可灌溉地（亩）	最大用水量（百立方）
1	400	600
2	600	800
3	300	375

各农场均可种植甜菜、棉花和高粱三种作物，各种作物的用水量、净收益及国家规定的该地区各种作物种植总面积最高限额如下表：

作物种类	种植限额（亩）	耗水量（百立方/亩）	净收益（元/亩）
甜菜	600	3	400
棉花	500	2	300
高粱	325	1	100

三个农场达成协议，他们的播种面积与其可灌溉面积相等，而各农场种何种作物并无限制.问如何制定各农场种植计划才能在上述限制条件下，使本地区的三个农场的总净收益最大.

对本题来说，由于数据少，可以不采用数组形式，而直接采用变量表示，建立模型如下：

设农场1种植的甜菜、棉花和高粱分别为 x_1, y_1, z_1 亩，农场2种植的甜菜、棉花和高粱分别为 x_2, y_2, z_2 亩，农场3种植的甜菜、棉花和高粱分别为 x_3, y_3, z_3 亩.

根据题目条件，可建立如下线性模型：

$$\max Z = 400(x_1 + x_2 + x_3) + 300(y_1 + y_2 + y_3) + 100(z_1 + z_2 + z_3)$$

$$s.t.\begin{cases} x_1 + x_2 + x_3 \le 600 \\ y_1 + y_2 + y_3 \le 500 \\ z_1 + z_2 + z_3 \le 325 \\ x_1 + y_1 + z_1 \le 400 \\ x_2 + y_2 + z_2 \le 600 \\ x_3 + y_3 + z_3 \le 300 \\ 3x_1 + 2y_1 + z_1 \le 600 \\ 3x_2 + 2y_2 + z_2 \le 800 \\ 3x_3 + 2y_3 + z_3 \le 375 \\ x_1, x_2, x_3, y_1, y_2, y_3, z_1, z_2, z_3 \ge 0 \end{cases}$$

LINGO程序如下：

```
MODEL:
max=400*(x1+x2+x3)+300*(y1+y2+y3)+100*(z1+z2+z3);
x1+x2+x3<=600;
y1+y2+y3<=500;
z1+z2+z3<=325;
x1+y1+z1<=400;
x2+y2+z2<=600;
x3+y3+z3<=300;
3*x1+2*y1+z1<=600;
3*x2+2*y2+z2<=800;
3*x3+2*y3+z3<=375;
END
```

得到的解如下：

X1=133.3333,Y1=100,Z1=0;X2=0,Y2=400,Z2=0;
X3=125,Y3=0,Z3=0;

最大总净收益为**253333.3**元.

解

设农场1种植的甜菜、棉花和高粱分别为 x_{11}, x_{12}, x_{13} 亩，农场2种植的甜菜、棉花和高粱分别为 x_{21}, x_{22}, x_{23} 亩，农场3种植的甜菜、棉花和 高粱分别为 x_{31}, x_{32}, x_{33} 亩.

设三个农场可耕地分别为 $a_1 = 400, a_2 = 600, a_3 = 300$

其最大用水量分别为 $b_1 = 600,$ $b_2 = 800$, $b_3 = 375$

其甜菜、棉花和高粱的种植限额分别为

$$c_1 = 600, c_2 = 500, c_3 = 325$$

其耗水量分别为 $d_1=3, d_2=2,\ d_3=1$

其净收益分别为 $e_1=400, e_2=300, e_3=100$

根据题目条件，可建立如下线性模型：

$$\max Z=\sum_{j=1}^{3}(e_j\sum_{i=1}^{3}x_{ij})$$

$$s.t.\quad \begin{cases} \sum_{i=1}^{3}x_{ij}\le c_j & j=1,2,3 \\ \sum_{j=1}^{3}x_{ij}\le a_i & i=1,2,3 \\ \sum_{j=1}^{3}d_jx_{ij}\le b_i & i=1,2,3 \end{cases}$$

LINGO编程如下：

```
model:
sets:
place/1..3/:a,b;
kind/1..3/:c,d,e;
plan(place,kind):x;
endsets
data:
a=400,600,300;
b=600,800,375;
c=600,500,325;
d=3,2,1;
e=400,300,100;
enddata
max=@sum(kind(j):e(j)*@sum(place(i):x(i,j)));
@for(kind(j):@sum(place(i):x(i,j))<=c(j));
@for(place(i):@sum(kind(j):x(i,j))<=a(i));
@for(place(i):@sum(kind(j):d(j)*x(i,j))<=b(i));
end
```

得到结果如下：

X(1,1)=133.3333,X(1,2)=100,X(1,3)=0
X(2,1)=0,X(2,2)=400,X(2,3)=0
X(3,1)=125,X(3,2)=0,X(3,3)=0

最大总净收益为**253333.3**元.

7. 有五项设计任务可供选择.各项设计任务的预期完成时间分别为3,8,5,4,10（周）,设计报酬分别为7,17,11,9,21（万元）.设计任务只能一项一项地进行，总的期限为20周.

选择任务时必须满足下面要求：

(1) 至少完成3项设计任务.

(2) 若选择任务1，必须同时选择任务2.

(3) 任务3和任务4不能同时选择.

应当选择哪些任务，才能使总的设计报酬最大？

解

这是一个0-1整数规划问题.

设0-1变量 x_i 如下：

$$x_i = \begin{cases} 0 & \text{第}i\text{项设计任务未选上} \\ 1 & \text{第}i\text{项设计任务被选上} \end{cases}$$

设各项设计任务的完成时间为$t_i (i = 1, 2, \cdots, 5)$表示，设计报酬为$m_i (i = 1, 2, \cdots, 5)$表示.

则容易得到目标函数：

$$\max Z = \sum_{i=1}^{5} m_i x_i$$

根据题目要求分别列出约束条件如下：

总期限为20周，则约束条件为$\sum_{i=1}^{5} t_i x_i \le 20$

至少完成3项设计任务，则$\sum_{i=1}^{5} x_i \ge 3$

若选择任务1，必须同时选择任务2，则$x_2 \ge x_1$

任务3和任务4不能同时选择，则$x_3 + x_4 \le 1$

数学模型：

$$\max Z = \sum_{i=1}^{5} m_i x_i$$

$$s.t.\quad \begin{cases} \sum_{i=1}^{5} t_i x_i \le 20 \\ \sum_{i=1}^{5} x_i \ge 3 \\ x_2 \ge x_1 \\ x_3 + x_4 \le 1 \\ x_i = 0\text{或}1,\ i = 1,2,\cdots,5 \end{cases}$$

LINGO程序如下：

```
MODEL:
SETS:
mat/1..5/:m,t,x;
ENDSETS
DATA:
m=7,17,11,9,21; !定义报酬数组;
t=3,8,5,4,10;   !定义完成时间;
ENDDATA
max=@SUM(mat(i):m(i)*x(i)); !定义目标函数;
@SUM(mat(i):t(i)*x(i))<=20; !期限约束 ;
@SUM(mat(i):x(i))>=3;   !至少完成3项任务;
x(2)>=x(1);           !若选择任务1，必须同时选择任务2;
x(3)+x(4)<=1;         !任务3和任务4不能同时选择;
@FOR(mat(i):@BIN(x(i)));    !使各变量为0-1变量;
END
```

解为：x(1)=1,x(2)=1,x(3)=1,x(4)=0,x(5)=0.

选择设计任务1,2,3，最大报酬为35万元.

8. 某企业和用户签定了设备交货合同，已知该企业各季度的生产能力、每台设备的生产成本和每季度末的交货量见下表，若生产出的设备当季度不交货，每台设备每季度需要支付保管费0.1万元，试问在遵守合同的条件下，企业应如何安排生产计划，才能使年消耗费用最低？

季度	工厂生产能力（台）	交货量（台）	每台设备生产成本（万元）
1	25	15	12.0
2	35	20	11.0
3	30	25	11.5
4	20	20	12.5

解法1

设第 i 季度生产 x_i 台，库存 y_i 台 ,$i=1,2,3,4$ 第 i 季度生产能力用 p_i 表示，交货量用 d_i 表示，每台设备生产成本用 c_i 表示.则建立目标函数为：

$$\min Z=\sum_{i=1}^{4}(c_i x_i+0.1y_i)$$

$$s.t.\begin{cases} x_i\le p_i & i=1,2,3,4 \\ y_i=y_{i-1}+x_i-d_i & i=2,3,4 \\ y_1=x_1-d_1 & \\ x_i\ge 0, y_i\ge 0 & \end{cases}$$

LINGO程序如下：

```
MODEL:
SETS:
QUART/1..4/:x,y,p,d,c;
ENDSETS
DATA: !指定数据;
p=25,35,30,20;
d=15,20,25,20;
c=12.0,11.0,11.5,12.5;
ENDDATA
min=@sum(QUART(i):c(i)*x(i)+0.1*y(i)); !目标函数;
@FOR(QUART(i):x(i)<=p(i)); !生产能力限制;
@FOR(QUART(i)|i#GT#1:y(i)=y(i-1)+x(i)-d(i));
y(1)=x(1)-d(1);
end
```

得到的结果如下：

x1=15，x2=35，x3=30，x4=0；y1=0，y2=15，y3=20，y4=0.

年消耗最小费用为913.5万元.

解法2

设x_{ij}第i季度生产第j季度交货的台数，第i季度生产能力用p_i表示，交货量用d_i表示。

由于生产能力的限制，需要满足下面条件：

$$\sum_{j=i}^{4} x_{ij} \le p_i \quad i=1,2,3,4$$

根据交货量的规定，应满足如下条件：

$$\sum_{i=1}^{j} x_{ij} = d_j \quad j=1,2,3,4$$

第i季度生产第j季度交货的每台设备所消耗的费用c_{ij}，应等于生产成本加上保管维护费用之和，其值如下表：

生产季度 \ 交货季度	1	2	3	4
1	12.0	12.1	12.2	12.3
2		11.0	11.1	11.2
3			11.5	11.6
4				12.5

则该模型表示如下：

$$\min Z=\sum_{i=1}^{4}\sum_{j=i}^{4}c_{ij}x_{ij}$$

$$s.t.\quad\begin{cases}\sum_{j=i}^{4}x_{ij}\le p_i & i=1,2,3,4\\ \sum_{i=1}^{j}x_{ij}=d_j & j=1,2,3,4\\ x_{ij}\ge 0 & i,j=1,2,3,4\end{cases}$$

LINGO程序如下：

```
MODEL:
SETS:
QUART/1..4/:p,d;
LINK(QUART,QUART)|&1#LE#&2:x,c; !只取上三角阵;
ENDSETS
DATA: !指定数据;
p=25,35,30,20;
d=15,20,25,20;
c=12.0  12.1  12.2   12.3
        11.0  11.1   11.2
              11.5   11.6
                     12.5;
ENDDATA
MIN=@SUM(LINK:c*x); !目标函数;
@FOR(QUART(i):@SUM(QUART(j)|j#GE#i:x(i,j))<=p(i)); !生产能力限制;
@FOR(QUART(j):@SUM(QUART(i)|i#LE#j:x(i,j))=d(j)); !交货合同限制;
End
```

得到的结果如下：

X(1,1)=15，X(1,2)=0，X(1,3)=0，X(1,4)=0；X(2,2)=20，X(2,3)=0，X(2,4)=15， X(3,3)=25, X(3,4)=5；X(4,4)=0.

年消耗最小费用为**913.5**万元.

可以看出，第1季度生产量为15台，第2季度生产量为35台，第3季度生产量为30台，第4季度生产量为0台，与前面方法得到的结果一样.其最小费用也一样.

9. 一奶制品加工厂用牛奶生产A_1,A_2两种奶制品，1桶牛奶可以在甲车间用12小时加工成3公斤A_1，或者在乙车间用8小时加工成4公斤A_2。根据市场需求，生产的A_1,A_2全部能售出，且每公斤A_1获利24元，每公斤A_2获利16元。现在加工厂每天能得到50桶牛奶的供应，每天正式工人总的劳动时间480小时，并且甲车间每天至多能加工100公斤A_1，乙车间的加工能力没有限制。

试为该厂制订一个生产计划，使每天获利最大，

并进一步讨论以下3个附加问题：

- 1） 若用35元可以买到1桶牛奶，应否作这项投资？
- 若投资，每天最多购买多少桶牛奶？
- 2） 若可以聘用临时工人以增加劳动时间，付给临时工人的工
- 资最多是每小时几元？

3） 由于市场需求变化，每公斤A_1的获利增加到30元，应否改变生产计划？

```
model:
max=24*3*x1+16*4*x2;
x1+x2<=50;
12*x1+8*x2<=480;
3*x1<=100;
end
```

所以，最优解为$x1=20,x2=30$,最优值z=3360,即用20桶牛奶生产A1,30桶牛奶生产A2，可最大获利3360元.

（1）用35元可以买到1桶牛奶，低于其对偶价格，故应该做这项投资；
（2）聘用临时工人来增加劳动时间，付给的工资低于劳动时间的对偶价格才可以增加利润，所以工资最多是每小时2元；
（3）若每公斤A_1的获利增加到30元，则x1系数变为90，在允许范围内，所以不应改变生产计划，但最优值变为90*20+64*30=3720。

10. 某公司有6个建筑工地要开工，每个工地的位置（用平面坐标系a，b表示，距离单位：km）及水泥日用量d(t)由下表给出．目前有两个临时料场位于A(5,1)，B(2,7)，日储量各有20t．假设从料场到工地之间均有直线道路相连。

试制定每天的供应计划，即从A，B两料场分别向各工地运送多少水泥，可使总的吨千米数最小。

工地位置（a，b）及水泥日用量d

	1	2	3	4	5	6
a	1.25	8.75	0.5	5.75	3	7.25
b	1.25	0.75	4.75	5	6.5	7.75
d	3	5	4	7	6	11

建立模型

记工地的位置为(a_i,b_i)，水泥日用量为$d_i, i=1,\cdots 6$；料场位置为(x_j,y_j)，日储量为 $e_j, j=1,2$；从料场j 向工地 i 的运送量为c_{ij} 。

$$\text{MIN} \quad f=\sum_{j=1}^{2}\sum_{i=1}^{6}c_{ij}\sqrt{\left(x_j-a_i\right)^2+\left(y_j-b_i\right)^2} \qquad (1)$$

$$\text{s.t.} \quad \sum_{j=1}^{2}c_{ij}=d_i,\quad i=1,2,\cdots,6 \qquad (2)$$

$$\sum_{i=1}^{6}c_{ij}\le e_j,\quad j=1,2 \qquad (3)$$

10’. 若丢弃A, B两料场，再新建两个料场分别向各工地运送水泥，问两个新的料场应建在何处，使总的吨公里数最小。

本例中集合的概念

利用集合的概念，可以定义需求点**DEMAND**和供应点**SUPPLY**两个集合，分别有**6**个和**2**个元素**(**下标**)**。

集合的属性相当于以集合的元素为下标的数组。这里的 c_{ij} 相当于二维数组。它的两个下标分别来自集合**DEMAND**和**SUPPLY**，因此可以定义一个由二元对组成的新的集合，然后将 c_{ij} 定义成这个新集合的属性。

link(demand,supply):c;

程序: eg3.lg4

定义了三个集合，其中LINK在前两个集合DEMAND 和SUPPLY的基础上定义

LINGO Model -

```
model:
sets:
demand/1..6/:a,b,d;
supply/1..2/:x,y,e;
link(demand,supply):c;
endsets
data:
a=1.25,8.75,0.5,5.75,3,7.25;
b=1.25,0.75,4.75,5,6.5,7.75;
d=3,5,4,7,6,11;
e=20,20;
enddata
```

表示集合LINK中的元素就是集合DEMAND和SUPPLY的元素组合成的有序二元组，

从数学上看LINK是DEMAND 和SUPPLY的笛卡儿积，也就是说

LINK={（S，T）|SDEMAND，TSUPPLY}

因此，其属性C也就是一个6*2的矩阵（或者说是含有12个元素的二维数组）。

LINGO建模语言也称为矩阵生成器（MATRIX GENERATOR）。类似DEMAND 和SUPPLY直接把元素列举出来的集合，称为基本集合(primary set),而把LINK这种基于其它集合而派生出来的二维或多维集合称为派生集合(derived set)。由于是DEMAND 和SUPPLY生成了派生集合LINK，所以DEMAND 和SUPPLY 称为LINK的父集合。

同理，数据段中对常数数组A，B的赋值语句也可以写成

A, B=1.25 1.25 8.75 0.75 0.5 4.75 5.75 5 3 6.5 7.25 7.75;

LINGO Model

```
model:
sets:
demand/1..6/:a,b,d
supply/1..2/:x,y
link(demand,sup         c;
endsets
data:
a=1.25,8.75,0.5,5.75,3,7.25;
b=1.25,0.75,4.75,5,6.5,7.75;
d=3,5,4,7,6,11;
e=20,20;
enddata
min=@sum(link(i,j):c(i,j)*((x(j)       ^2+(y(j)-b(i))^2)^(1/2));
@for(demand(i):@sum(supply(       i,j))=d(i));
@for(supply(i):@sum(dem     (j):c(j,i))<=e(i));
@for(supply:@free(x),@free(y));
init:
x,y=5,1,2,7;
endinit
end
```

LINGO对数据是按列赋值的

语句的实际赋值顺序是

X=(5,2), Y=(1,7), 而不是

X=(5,1), Y=(2,7)

等价写法:

“X=5,2； Y=1,7； ”

初始段

LINGO Model - eg31

```
model:
sets:
demand/1..6/:a,b,d;
supply/1..2/:x,y,e;
link(demand,supply):c;
endsets
data:
a=1.25,8.75,0.5,5.75,3,7.25;
b=1.25,0.75,4.75,5,6.5,7.75;
d=3,5,4,7,6,11;
e=20,20;
enddata
min=@sum(link(i,j):c(i,j)*((x(j)-a(i))^2+(y(j)-b(i))^2)^(1/2));
@for(demand(i):@sum(supply(j):c(i,j))=d(i));
@for(supply(i):@sum(demand(j):c(j,i))<=e(i));
@for(supply:@free(x);@free(y));
init:
x,y=5,1,2,7;
endinit
end
```

定义目标和约束，与前例的方法是类似(这里包含了派生集合)，请特别注意进一步体会集合函数@SUM和@FOR的用法。

由于新建料场的位置理论上讲可以是任意的，所以在约束的最后(模型的“END”语句上面的一行)用@free函数取消了变量X、Y的非负限制

谢谢！